

Procrastination

A proof engineering technique

HAVE YOU SEEN THOSE PROCRASTINATORS THE MONKS USE?
WONDERFUL THINGS. THEY CAN MOVE TIME, STORE IT, STRETCH IT...
QUITE INGENIOUS.

— Terry Pratchett, *Thief of Time*

ABSTRACT

We present a small Coq library for collecting side conditions and deferring their proof.

1 INTRODUCTION

What are the different ways of proving an assertion of the form “ $\exists x, P\ x$ ”? From the perspective of the final proof term, there is only one: providing a witness. However, from a “proof engineering” perspective, Coq enables a wider range of possible approaches thanks to its powerful feature of “evars” [5, §2.11].

We describe existing approaches for proving such a goal, and propose a new one, supported by a small library dubbed *Procrastination* [3], which itself relies on evars. We believe it captures a useful proof pattern, where one wishes to first continue with the proof of $P\ x$, in order to discover which constraints x must satisfy, before providing a value for x and proving the side conditions.

We used this library with success in our work on verifying the asymptotic complexity of programs [4].

2 PROVING EXISTENTIAL QUANTIFICATIONS IN COQ

Manually guessing a witness. The most straightforward option to prove a Coq goal “ $\exists x, P\ x$ ” is to guess and provide upfront a witness, using the `exists` tactic. This is often a good way of providing external knowledge about how the proof should continue.

Delaying the instantiation using evars. However this falls short if the only intuition is that x should be “whatever makes the proof work”. A trial-and-error process can be attempted, where the user alternatively adjusts the witness and the proof until it passes. This does not lead to maintainable proof scripts: in the end, only a “magic” witness remains; it may be large and duplicate information; it may also have to change in the future, following updates to the rest of the proof.

To allow delaying the instantiation of x , Coq provides a versatile feature: evars. An evvar corresponds to a “hole” in the proof, and only exists during the pre-typing phase – it will not appear in the proof term, and only helps elaborating it. Using the `eexists` tactic on the goal “ $\exists x, P\ x$ ” turns it into “ $P\ ?x$ ”, introducing a fresh evvar $?x$. Later, it can be discovered that this hole should be (definitionally) equal to some term: the evvar then gets instantiated. This is generally performed automatically by Coq tactics through unification.

“Big enough” existentials. This technique again falls short if continuing with the proof of $P\ ?x$ does not exhibit a term that $?x$ must be *equal* to, but e.g. *lower bounds* on $?x$. Cohen [1, §5.1.4] shows that it is a common situation in analysis, where a proof is established by picking a “big enough x ”. During the proof, inequalities of the form “ $a \leq x$ ”, “ $b \leq x$ ”, ... are all trivially satisfied provided that “ x is big enough”.

Cohen formalizes this seemingly fuzzy reasoning in a small tactic library. On a “ $\exists x, P\ x$ ” goal, a tactic instantiates x with the iterated maximum of a list of terms, this list being initially an evvar. It is then possible to progressively populate this list, automatically discharging subgoals of the form “ $a \leq x$ ”.

3 THE PROCRASTINATION LIBRARY

We are interested in the situation where arbitrary side conditions about x may be encountered – e.g. inequalities of both the form “ $a \leq x$ ” and “ $x \leq b$ ”. We might also be trying to infer not an arithmetic value but a function, e.g. f satisfying “ $\forall n, f(n) \geq 1 + f(n-1)$ ”. In this situation, it seems difficult to guess the shape of x or f beforehand (e.g. as an iterated maximum).

We introduce a small Coq library named *Procrastination*, which generalizes the ideas of Cohen, and allows collecting arbitrary side conditions (to be proved later) about zero, one or several variables. It is more general than “big enough” in the sense that it can deal with arbitrary side conditions, but it does not try to solve these side conditions: this is typically done in a second time by a dedicated procedure (e.g. for “big enough” this procedure would take the running maximum of the lower-bounds collected).

In other words, the library enforces a pattern where the proof is separated in two phases: first side conditions are collected about some variables, then they are solved and the variables are instantiated. This is crucial in limiting the proliferation of evars, which tend to produce brittle proofs because they can might get incorrectly instantiated by unrelated tactics.¹

The implementation provides reusable Ltac components, with the hope that they can help applying this technique even in more exotic scenarios. The code is freely available online [3].

4 EXAMPLE: SOLVING RECURRENCE (IN)EQUATIONS

Our initial motivation for this work was analyzing the time complexity of programs, where one has to find a solution for a set of inequations about a program’s execution time. One approach is using Cormen *et al.*’s substitution method [2, §4]. The idea is to guess a parameterized *shape* for the solution; substitute this shape into the goal; gather a set of constraints the parameters must satisfy for the goal to hold; finally, show that these constraints are indeed satisfiable.

We demonstrate this strategy using *Procrastination* on cost inequations derived from a binary search program. We “guess” that the solution should be a monotonic function f with shape λn . if $n = 0$ then c else $(a \log n + b)$; where a , b and c are parameters, about which we wish to gradually accumulate a set of constraints.

```
Goal  $\exists (f: \text{nat} \rightarrow \text{nat}),$ 
   $1 \leq f\ 0 \wedge$ 
  monotonic  $f \wedge$ 
   $\forall n, 0 < n \rightarrow 2 + f\ (n/2) \leq f\ n.$ 
Proof.
  begin procrastination assuming a b c. (1)
  exists (fun n  $\Rightarrow$  if zero n then c
    else  $a * \log_2 n + b$ ).
  repeat split.
  { simpl. (2) procrastinate. (3) }
  { ... (* c  $\leq b$  *) procrastinate. ... }
  { intros.
    ... (*  $2 + c \leq b$  *) procrastinate.
    ... (*  $2 \leq a$  *) procrastinate. }
  end procrastination.
  (4) eomega.
Qed.
```

$a: ?x$ $b: ?x0$ $c: ?x1$ $g: \text{Group } (?G\ a\ b\ c)$			(1)
$\exists f: \text{nat} \rightarrow \text{nat},$ $1 \leq f\ 0 \wedge \text{monotonic } f \wedge \forall n, 0 < n \rightarrow 2 + f\ (n/2) \leq f\ n$			
$a, b, c: \text{nat}$ $g: \text{Group } (?G\ a\ b\ c)$	(2)	$a, b, c: \text{nat}$ $g: \text{Group } (1 \leq c \wedge ?Goal0)$	(3)
$1 \leq c$		Subgoal solved	
$\exists a\ b\ c, 1 \leq c \wedge c \leq b \wedge 2 + c \leq b \wedge 2 \leq a$			(4)

The script starts with `begin procrastination assuming a b c` which introduces the parameters, as well as a “group” g , using an `evvar` to collect the constraints (1). Note that a , b and c are not evvars themselves, instead they appear as abstract variables. Note also that the type of these variables is inferred to be `nat` only further on in the proof script. After instantiating the function with the proposed solution, the first inequality simplifies to a condition on c (2). This condition is stored (in g) for later retrieval, using the `procrastinate` tactic, which discharges the subgoal (3). Similarly, the other two inequalities yield side conditions about a , b and c that can be procrastinated. Finally, `end procrastination` retrieves the collected conditions, which the user must then prove. Here, this is done by an ad hoc user-defined tactic, `eomega`.

Observe that the final proof script never mentions the actual witnesses: we expect the proof to be robust in the face of minor changes in the cost inequations, as long as the *shape* of the solution remains valid.

¹For example, we noticed that `ring_simplify` is particularly careless (or sometimes crashes) when the goal contains evvars.

5 EXAMPLE: SIMPLIFYING RECURRENCE (IN)EQUATIONS

An other application of the approach, also coming from our work on complexity analysis, is to simplify (in)equations about a function. The inequalities constituting the goal of the example in Section 4 have in fact already been simplified. In practice, analyzing the complexity of a program yields a set of constraints C and a goal $\exists f, C(f)$. C has typically been elaborated automatically from the program, and one wants to simplify the goal into $\exists f, C'(f)$, where C' is a simpler set of constraints.

In this example we show that we can use *Procrastination* to collect constraints not only about natural numbers (as in the previous example), but also about functions. Starting from constraints (on the program execution time) that are close to what an automated program analysis would produce, we are able to simplify them to finally obtain the simpler constraints of the previous example.

```
Goal  $\exists (f: \text{nat} \rightarrow \text{nat}),$ 
   $\forall n,$ 
     $1 + (\text{if } \text{zerop } n \text{ then } 0$ 
       $\text{else } 1 + \max (f (n/2)) (f (n - (n/2) - 1)))$ 
     $\leq f n.$ 
Proof.
  begin procrastination assuming f. (1) exists f.
  intro n. destruct (zerop n) as [Hn].
  { subst n. simpl. (2) procrastinate. }
  { rewrite max_1; swap 1 2.
    { (3) procrastinate M: (monotonic f). (4) ... }
    { ... (5) revert n Hn. (6) procrastinate. } }
  end procrastination.
(7) ...
Qed.
```

$f: ?x$ $g: \text{Group } (?G f)$	(1)
<hr/>	
$\exists (f0: \text{nat} \rightarrow \text{nat}),$ $\forall n, 1 + (\text{if } \text{zerop } n \text{ then } 0$ $\text{else } 1 + \max (f0 (n/2)) (f0 (n - (n/2) - 1)))$ $\leq f0 n$	
<hr/>	
$f: \text{nat} \rightarrow \text{nat}$ $g: \text{Group } (?G f)$	(2)
<hr/>	
$1 \leq f 0$	

$f: \text{nat} \rightarrow \text{nat}$ $g: \text{Group } (1 \leq f 0 \wedge ?\text{Goal}0)$ $n: \text{nat}$ $H: 0 < n$	(3)
<hr/>	
$f (n - n / 2 - 1) \leq f (n / 2)$	

$f: \text{nat} \rightarrow \text{nat}$ $g: \text{Group } (1 \leq f 0 \wedge \text{monotonic } f \wedge ?\text{Goal}0)$ $n: \text{nat}$ $H: 0 < n$ $M: \text{monotonic } f$	(4)
<hr/>	
$f (n - n / 2 - 1) \leq f (n / 2)$	

$f: \text{nat} \rightarrow \text{nat}$ $g: \text{Group } (1 \leq f 0 \wedge \text{monotonic } f \wedge ?\text{Goal}0)$ $n: \text{nat}$ $H: 0 < n$	(5)
<hr/>	
$2 + f (n / 2) \leq f n$	

$f: \text{nat} \rightarrow \text{nat}$ $g: \text{Group } (1 \leq f 0 \wedge \text{monotonic } f \wedge ?\text{Goal}0)$	(6)
<hr/>	
$\forall n, 0 < n \rightarrow 2 + f (n / 2) \leq f n$	

$\exists f: \text{nat} \rightarrow \text{nat},$ $1 \leq f 0 \wedge \text{monotonic } f \wedge$ $\forall n, 0 < n \rightarrow 2 + f (n/2) \leq f n$	(7)
--	-----

The first step of the proof is to start collecting constraints about f , using *begin procrastination assuming f* (1). This gives us a new identifier f , that we can use right away as a witness for the existential quantification at the head of the goal. Case splitting first allows extracting the inequation for the base case (2). Some work can be done in the recursive case (3), if we additionally assume that f is monotonic (4). In that case the $\max \dots$ can be simplified to its first argument (5). The inequation at (5) is the simplified inequation for the recursive case. However, it is not possible to call *procrastinate* at this point: the goal depends on the local variable n , while the set of collected constraints can only depend on f . The solution is to generalize the goal to any positive n (6), by moving n and Hn from the local context to the goal. Finishing the procrastination yields the set of simplified constraints, which corresponds to the initial goal of the previous example (7).

Here again, the point of using *Procrastination* is to impose a *proof discipline* that improves the robustness of the proof script. In this example, we assumed that the goal has been produced by a mechanized process. We would like the proof script to be maintainable, with respect to (non-fundamental) changes in this mechanized process. *Procrastination* is used a first time to extract a set of simplified constraints, then a second time (as detailed in Section 4) to find a solution for the simplified constraints, without relying on the exact values in these constraints. By explicitly splitting the proof into two parts that operate at two different levels of abstraction, we can hope that if the structure of the initial goal changes, only the first half of the proof will need to be updated.

6 PROCRASTINATION AND THE PROOF CONTEXT

As illustrated previously, one can think of *Procrastination* as a way of incrementally collecting some side-conditions P, Q, R about some variables a, b, c in order to eventually produce the goal: $\exists a b c, P \wedge Q \wedge R$.

There are two key aspects to this process.

Side-conditions are collected and gathered under a common context. Specifically, it is the context at the point of calling “begin procrastination assuming $a b c$ ”, extended with the names a, b and c . Only side-conditions P, Q and R that make sense in this context can be deferred using procrastination.

More generally, to each “procrastination block” (beginning with `begin procrastination` and ending with `end procrastination`) corresponds a proof context in which the side-conditions will be gathered. This is in fact materialized by an assumption of type “Group (...)” added to the context by `begin procrastination`. This assumption stores the current collection of side-conditions, and can be named explicitly by the user.

Requiring all side-conditions to be expressed in the same context allows the library to collect them as a conjunction in a single final subgoal $\exists a b c, P \wedge Q \wedge R$. This is crucial, since the motivation is to find values for a, b, c by inspecting the side-conditions P, Q, R altogether (using a tactic or manually).

When collecting side-conditions, variables a, b, c are rigid. Because the end goal is to find an instantiation for a, b and c , one could think of first introducing evvars for these, and then try to collect side-conditions. However, this quickly becomes unwieldy, as the evvar for collecting side-conditions will depend itself on evvars for a, b and c . This is in general tricky to handle, the main risk being that a, b or c get accidentally instantiated in undesired ways while trying to collect side-conditions about them.

Procrastination imposes a two-step process, where side-conditions are first collected with respect to “rigid” names for a, b and c . This limits the number of evvars involved to a single one in “Group (...)”, which is only manipulated by the library tactics. After the side-conditions have been collected, the user is free to introduce evvars for a, b and c , if needed to prove the goal $\exists a b c, P \wedge Q \wedge R$.

7 NESTING PROCRASTINATION BLOCKS

Interestingly, it is possible to nest procrastinate blocks in an arbitrary way. In some sense, collecting side-conditions is done in a “well-scoped” manner. For example, it is possible to open a second procrastinate block while being in a procrastinate block. Then, the nested block will have a larger context than the enclosing block. When using `procrastinate` on a side-condition, the user can control in which Group (and therefore in which context) the side-condition gets collected.

8 IMPLEMENTING PROCRASTINATION: CORE IDEAS

In essence, the implementation of *Procrastination* is extremely simple. As its core, “begin procrastination assuming a ” is nothing more than “`eapply procrastinate_1`”, where `procrastinate_1` is the following tautological lemma:

```
Lemma procrastinate_1 : ∀ A (P: Prop) (Q: A → Prop),
  (∀ a, Q a → P) →
  (∃ a, Q a) →
  P.
```

This lemma (and the rest of the library) comes from the combination of two main ideas, described below.

8.1 Collecting side-conditions

Independently from reasoning on additional parameters a, b, c, \dots , the first idea is the one of incrementally collecting side-conditions while progressing through a proof.

Introducing additional facts in the middle of a proof corresponds to the usual *Cut* rule (or cut tactic). *Cut* can be stated as the following Coq lemma “on a goal P , we can assume some facts Q if we prove them later”:

```
Lemma cut_lemma :  $\forall (Q P : \text{Prop}), (Q \rightarrow P) \rightarrow Q \rightarrow P$ .
```

When using cut (the tactic) or when doing apply cut_lemma, the user needs to state upfront which the additional facts Q that are being assumed (and which proof is deferred for later) – e.g. apply (cut_lemma (foo \wedge bar)).

Now, interestingly, one can also delay giving Q , by calling “eapply cut_lemma”. This makes use of Coq’s evvars mechanism, and introduces an evvar for Q , since it hasn’t been determined yet.

```
Goal P.  
  eapply cut_lemma.  
  - intro H.  
    (* H: ?Q  $\vdash$  P *)  
  - (*  $\vdash$  ?Q *)
```

In the first subgoal, we get an assumption H which type is an evvar. Should P yield a subgoal P' that we may want to prove later, P' can simply be discharged by applying H . This will cause the evvar $?Q$ to get instantiated with P' – requiring P' to be proved in the second subgoal spawned by eapply cut_lemma as a side effect.

```
Goal P.  
  eapply cut_lemma.  
  - intro H.  
    (* H: ?Q  $\vdash$  P *)  
    ...  
    (* H: ?Q  $\vdash$  P' *)  
    now apply H.  
    (* H: P' *)  
  - (*  $\vdash$  P' *)
```

This trick can be made a bit more useful by using H to discharge not only one subgoal, but several. This can be achieved (e.g. using a small amount of Ltac) by instantiating H into an iterated conjunction of already discharged subgoals, and an evvar for future ones. For instance:

```
Goal P.  
  eapply cut_lemma.  
  - intro H. (* H: ?Q  $\vdash$  P *) ...  
    (* H: ?Q  $\vdash$  P' *) now eapply (proj1 H). ...  
    (* H: P'  $\wedge$  ?Q  $\vdash$  P'' *) now eapply (proj1 (proj2 H)). ...  
    (* H: P'  $\wedge$  P''  $\wedge$  ?Q  $\vdash$  ... *)  
  - (*  $\vdash$  P'  $\wedge$  P''  $\wedge$  ?Q *)
```

This is in fact precisely how the procrastinate tactic is implemented.

8.2 Dealing with existentials in the goal: avoiding the proliferation of evvars

Let us now consider the more specific case where P is of the form “ $\exists x, P\ x$ ”. In order to collect side-conditions that may talk about x , one possibility is to first call eexists, introducing an evvar for x , and then use the technique described previously.

```
Goal  $\exists x, P\ x$ .  
  eexists. (*  $\vdash$  P ?x *)
```

As mentioned earlier, in our experience this does not scale up very well. The risk is that $?x$ get instantiated by mistake while simply trying to collect side-conditions about it, or while progressing through the proof of $P ?x$.

Therefore, the second main idea is to use variants of the cut lemma for collecting side-conditions “under existential quantifications”. For example, we can easily prove the following lemma:

```
Lemma cut_ex1_lemma:  $\forall A (P Q : A \rightarrow \text{Prop}),$ 
  ( $\forall x, Q x \rightarrow P x$ )  $\rightarrow$ 
  ( $\exists x, Q x$ )  $\rightarrow$ 
   $\exists x, P x$ .
```

By doing eaply cut_ex1_lemma, we can collect side-conditions about x in $?Q x$, without introducing an evar for x . All in all, all this does is massaging the proof so that it is setup more conveniently.

Finally, the lemma above can be generalized to support an arbitrary number of quantified variables (including 0, in which case this corresponds to the previous subsection). This is in essence how begin procrastination is implemented.

9 TACTICS REFERENCE

- `begin procrastination [group g] [assuming a b...]`
Start a “procrastination” block. If `group g` is provided, give name `g` to the hypothesis of type `Group` introduced. If `assuming a b...` (where `a, b, ...` are one or several user-provided names) is provided, also introduce in the context abstract variables of the same names.
- `procrastinate [group g]`
Discharge the current sub-goal, deferring it for later. If `group g` is provided, store it in the hypothesis `g` which must be of type `Group` ... If `group g` is not provided, the tactic picks the hypothesis of type `Group` ... which has been introduced last (if there are several).
The sub-goal being differed has to make sense in the context of the `Group` it is stored in (which is the context of the `begin procrastination` that introduced the `Group`). Otherwise Coq will raise an error involving the context of the evar in the `Group`.
- `procrastinate [intropat] : H [group g]`
`procrastinate name: H group g` is a shorthand for `assert (name: H) by (procrastination group g)`.
In other words, it adds an hypothesis of type `H` to the context, and defers its proof. `group g` is optional, with the same behavior as `procrastinate`.
- `end procrastination`
To be called on a `end procrastination` goal. It does some cleanup and gives back the variables and side-goals that have been procrastinated.
- `already procrastinated [group g]`
`already procrastinated group g` proves a sub-goal if it has already been procrastinated before in the `group g`.
`group g` is optional, with the same behavior as in `procrastinate`.
- `already procrastinated [intropat] : H [group g]`
`already procrastinated name : H group g` adds only one hypothesis from `g : Group ...` to the context, of type `H`, and names it `name`. This is a more explicit version of `already procrastinated`. E.g. with `g : Group (P /\ Q /\ ?G)`, `already procrastinated H: Q group g` introduces an hypothesis `H` of type `Q`.
`group g` is optional, with the same behavior as in `procrastinate`.
- `with procrastinated [group g] [do tac]`

- `with procrastinated` adds all hypothesis corresponding to the already procrastinated facts stored in `g`. E.g. if `g` has type `Group (P /\ Q /\ ?G)`, `already procrastinated group g` adds an hypothesis of type `P` and an hypothesis of type `Q`.
This is useful since tactics (e.g. `omega`) are not necessarily able to peek into the `Group _` hypothesis (and it is not always desirable anyway).
- `with procrastinated do tac` allows iterating on already procrastinated facts. `tac` must be a tactic `lambda` taking an hypothesis name as input. It will be called on each term that has already been procrastinated in group `g`.
`group g` is optional, with the same behavior as in `procrastinate`.

9.1 “defer” variants of the tactics

Variants of the tactics are also provided that use “defer” in place of “procrastinate”, “defer” being easier to type.

- `begin deferring` is an alias for `begin procrastination`
- `defer` is an alias for `procrastinate`
- `end deferring` is an alias for `end procrastination`
- `already deferred` is an alias for `already procrastinated`
- `with deferred` is an alias for `with procrastinated`

REFERENCES

- [1] Cohen, C.: [Formalized algebraic numbers: construction and first-order theory](#). Ph.D. thesis, École Polytechnique (2012)
- [2] Cormen, T.H., Leiserson, C.E., Rivest, R.L., Stein, C.: [Introduction to Algorithms \(Third Edition\)](#). MIT Press (2009)
- [3] Guéneau, A.: The procrastination library (Jul 2018), <https://github.com/Armael/coq-procrastination>
- [4] Guéneau, A., Charguéraud, A., Pottier, F.: [A fistful of dollars: Formalizing asymptotic complexity claims via deductive program verification](#). In: European Symposium on Programming (ESOP) (2018)
- [5] The Coq development team: [The Coq Proof Assistant](#) (2016)